# $Id: PKGBUILD 88687 2013-04-21 22:17:03Z heftig $
# Maintainer:  Ionut Biru <ibiru@archlinux.org>
# Contributor: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Hugo Doria <hugo@archlinux.org>
# Contributor: Sergej Chodarev <sergejx@centrum.sk>
# Contributor: zhuqin <zhuqin83@gmail.com>

pkgname=gedit-plugins
pkgver=3.8.1
pkgrel=1
pkgdesc="Plugins for gedit"
arch=(x86_64 i686)
license=(GPL)
url="http://live.gnome.org/GeditPlugins"
depends=(gedit python-dbus)
makedepends=(intltool gnome-doc-utils vte3)
optdepends=('gucharmap: for charmap plugin'
            'vte3: for embedded terminal')
options=('!libtool' '!emptydirs')
install=gedit-plugins.install
source=("http://ftp.gnome.org/pub/gnome/sources/$pkgname/${pkgver:0:3}/$pkgname-$pkgver.tar.xz")
sha256sums=('659d379f9edfbba2db37522991c734c5d50e03eb1f70985e978d93a718273141')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --sysconfdir=/etc --with-plugins=all \
      --disable-schemas-compile
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR=$pkgdir install
}

# vim:set ts=2 sw=2 et:
